
Lab 6: Clustered and Hierarchical Data

Overview

1. Inspect the BSI data

2. Add a family random effect to the BSI data 

3. Add a home scale random effect to the weight loss data

We look at two different data sets this time.  In the BSI data, adolescents are nested inside parent.  We have a longitudinal data within adolescents.  But shouldn’t children of the same parent be similar?  This should be accounted for in our models.  

The bsi has a total (global severity index, gsi) and several subscales.  Scales are the averages of several 0-4 items that make up each scale.  We realize that the scales are skewed, and take logs before analyzing.  

Lab Tasks

1. Inspect the BSI data

libname rel  "C:\Documents and Settings\rob\Desktop\temp"; *assumes dataset on desktop/temp;
options pageno=1 ls=90 ps=50;

proc contents data=rel.bsitotal;

run;

proc univariate data=rel.bsitotal;


var bsi_gsi bsi_dep bsi_anx bsi_soma;


histogram bsi_gsi bsi_dep bsi_anx bsi_soma;

run;

Initial inspection shows skewed responses.  We should take logs first.  We take logs base 2 because it’s a little easier to interpret.  Add a constant because zero is a common value.  Each constant is the smallest possible non-zero value for each subscale.  Also, we define two additional time variables to use in the upcoming model.  
data rel.bsitotal;


set rel.bsitotal;


lbsi_gsi = log2(bsi_gsi + 1/53);


lbsi_dep = log2(bsi_dep + 1/6);


lbsi_anx = log2(bsi_anx + 1/6);


lbsi_soma = log2(bsi_soma + 1/7);



month18 = max(0 , true_month - 18);



month36 = max(0 , true_month - 36);


label lbsi_anx='Log2(BSI Anxiety)'


lbsi_dep='Log2(BSI Depression)'


lbsi_gsi='Log2(BSI Global Severity Index)'


lbsi_soma='Log2(BSI Somatization)';

run;

proc contents data=rel.bsitotal;

run;

proc univariate data=rel.bsitotal;


var lbsi_gsi lbsi_dep lbsi_anx lbsi_soma;


histogram lbsi_gsi lbsi_dep lbsi_anx lbsi_soma;

run;

Inspection shows that the log global severity index (GSI) is much less skewed than before.  That’s great.  It’s still not normally distributed though.  The subscales are not worse, but not necessarily hugely better.  The large mass at zero is still there, and that cannot be fixed.  Still, when analyzing data, it is easier to do approximately the same thing to each response than to do something different to each one, and then have to explain it in your write-up.  

2. Fit the BSI GSI scale.  

First we fit without the parent effect.    

Predictors  include functions of time to allow two changes in slope giving a bent line model with knots (slope changes) at 18 and 36 months; gender; season.  

The covariance model is an ARMA(1,1) using the correct times, rounded to the nearest three months.  The fixed effects slopes use the exact times that observations were measured.  Season is a three group variable: winter is months 11, 12, 1, 2, spring 3—6, summer 7—10.  

Title1 '1 BSI no parent RE';

proc mixed data=rel.bsitotal noclprint covtest;

  class pid parent rounded3_true_month season gender;

  model lbsi_gsi = true_month month18 month36 season gender / s;

  repeated rounded3_true_month /sub=pid(parent) type=arma(1,1);

*SLOPE ESTIMATES;
  estimate 'slope before month 18' true_month 1;

  estimate 'slope between month 18 and 36' true_month 1 month18 1;

  estimate 'slope after month 36' true_month 1 month18 1 month36 1;

run;

TODO:  (1) Investigate the results:  (a) Where are the estimates of the slopes given?  Where are the estimated changes in the slopes given?  What are the slopes of lbsi_gsi over time, and how do they change at 18 and 36 months?  Are the slopes significantly different from zero?  Are the changes significantly different from zero?  


  (b) Describe the season effect, and is it significant?  


  (c) Ditto gender.  


  (d) Find the covariance parameters.  What is the residual variance equal to?  

TODO:  (2) Find the (REML) log likelihood value from this model.  

TODO:  (3) Add in the following random parent effect statement.  It goes right after the repeated statement above.  

  random intercept / type=un sub=parent;

and change the title, as well, for example to:

Title1 '2 BSI yes parent RE';

Best is to copy the above lines a second time, then make the modifications to new code.  That way we have both sets of code.  If there is a problem later, we can debug them separately.  

We must put parent into the class statement.  But it was already there in the previous analysis, which did not seem to need it.  Why?  If parent id was missing on some observations, then those observations must be omitted from both analyses in order to properly compare the two analyses.  Since we knew we would be adding in the parent random effect, and wanting to make a LRT comparison between the two models, we prepared ahead of time and put parent in the class statement of the first model.  

 TODO:  (4) Do the likelihood ratio test between these two models.  What is the test statistic, on how many degrees of freedom, and the conclusion?

TODO:  (5) Find the variance parameters for this model.  There is a new parameter here.  What is it, and what is its interpretation?  The new variance parameter plus the residual variance sum to equal almost exactly the residual variance from model 1.  Why?  

TODO:  (6)  Try some of the other BSI subscales.  Do you always conclude that you need a parent effect?  Is there always a season effect?  Is there always a gender effect?  Does time always have the same effect?   

3. Weight loss data.  

Now scale takes on a different meaning.  

TODO:  (7) Download the Weight Loss 2 data set.  This data set is in long format, and has weight as response, day is the day that the measurement was taken from the first visit, and ‘visit’ which is 1 for a home visit and 0 for a clinic visit.  

TODO:  (8) Do not use visit as a predictor.  Set up a RIAS model that is polynomial in day as the predictor.  I suggest defining d1 = day/10, as otherwise the polynomial models we will be fitting may cause SAS problems.  

TODO:  (9) Decide what order polynomial is needed to fit the data.  

TODO:  (10) Add visit as a fixed effect.  Now what order polynomial do you need to fit the data?  

TODO:  (11) Add visit as a random effect.  This can be done in two ways, as an independent random effect, or as a random effect that is correlated with the random intercept and slope already in the model.  Decide which is the better model: one of these models or the model without any visit random effect.  

TODO:  (12) Interpret both models, one where visit is correlated with the random intercept and slope, and where it is independent.  

TODO:  (13) The intervention actually started at the second visit, not the first visit.  Assuming no weight loss or gain between visits, the first and second visit should have the same mean weight.  After that point, weight may gain or decrease linearly (or by higher order polynomial).  Fix the predictors so that you can fit this model.  

